BONNES PRATIQUES VBA – Manipuler les graphiques (charts)

[bookmark: _Hlk35368646]
MANIPULER LES GRAPHIQUES (CHARTS) EN VBA



Présenter des données sous forme graphique nous permet d’exposer ces données de façon parlante et rapidement assimilable.
L’un des aspects des plus remarquables d’EXCEL réside dans sa capacité à présenter les données présentes dans un classeur sous forme de graphiques. Un ensemble très riche de fonctions permettent de gérer leurs présentations, leurs couleurs, leurs compositions…
Cette fiche « Bonnes pratiques VBA », sans couvrir le sujet in extenso, spécifie les objets majeurs qu’il est nécessaire de connaître pour manipuler les graphiques par programmation VBA.
(Les copies-écrans du présent document sont issues d’EXCEL version 365)

L’icône  [image: ] signale une remarque réclamant une attention particulière.

	
L’icône  [image: ] marque les commentaires en marge du sujet traité. Ils peuvent être ignorés sans inconvénient.




Table des matières

UTILISONS L’ENREGISTREUR DE MACROS	4
DISTINGONS LES 2 TYPES DE GRAPHIQUE	5
Bouclons sur les graphiques contenus dans un classeur	6
CREONS UN NOUVEAU GRAPHIQUE (CHART)	7
CREONS UN GRAPHIQUE INCORPORE	7
Méthode  ChartObjects.Add (Left, Top, Width, Height)	7
Méthode  Shapes.AddChart2 (style, XlChartType, Left, Top, Width, Height, NewLayout)	8
CREONS UN GRAPHIQUE ISOLE	9
Méthode  Charts.Add2	9
ILLUSTRONS PAR UN EXEMPLE	10
SPECIFIONS LES DEUX ZONES DE L’OBJET CHART	12
La propriété Chart.ChartArea	13
La propriété Chart.PlotArea	15
SPECIFIONS DES OBJETS MAJEURS DE L’OBJET CHART	17
L’objet ChartTitle	18
L’objet Legend	21
L’objet Axis	24
SPECIFIONS LA COLLECTION SERIESCOLLECTION	28
AJOUTONS DES SERIES DE DONNEES AU GRAPHIQUE	30
La méthode SeriesCollection.NewSeries	30
MODIFIONS LES CARACTERISTIQUES DES SERIES	31
Les objets Series.Values et Series.XValue	31
L’objet Series.Format	32
La propriété Series.Name	33
AJOUTONS UNE COURBE DE TENDANCE	34
Méthode Serie.TrendLines.Add(xlTrendLineType,…)	34
APPLIQUONS UN STYLE PREDEFINI AU GRAPHIQUE	35
La propriété Chart.Style (lStyle)	35


SPECIFIONS QUELQUES METHODES UTILES DE LA CLASSE CHART	36
EXPORTONS UN GRAPHIQUE EN PDF	36
Méthode Chart.ExportAsFixedFormat (Type, FileName,…)	36
EXPORTONS UN GRAPHIQUE AU FORMAT JPEG	36
Méthode Chart.Export (FileName…)	36
IMPRIMONS UN GRAPHIQUE	37
Méthode Chart.PrintOut (…)	37
ENREGISTRONS UN GRAPHIQUE EN TANT QUE MODELE	37
Méthode Chart.SaveChartTemplate(Filename)	37




[bookmark: _Toc46148468]UTILISONS L’ENREGISTREUR DE MACROS




 [image: ]Du fait de la richesse et de la diversité des méthodes et propriétés exposées par l’objet graphique (chart), nous gagnons beaucoup de temps en utilisant l’enregistreur de macros. 
Pour une bonne utilisation de l’enregistreur de macros, reportons-nous au document : 
https://www.excel-pratique.com/fr/telechargements/doc-vba/enregistreur-macro-excel-no467
Notons toutefois que, dans le cas de manipulations de graphique, l’enregistreur de macros :
· N’a pas la capacité de générer du code VBA pour toutes les manipulations que nous pouvons réaliser sur un graphique.
· Génère un code VBA souvent ‘bavard’ et non optimal.
Nous l’utiliserons principalement pour, dans un premier temps, découvrir les objets graphiques et, dans un second temps, nous corrigerons le code généré pour l’adapter et/ou le simplifier. 





[bookmark: _Toc46148469]DISTINGONS LES 2 TYPES DE GRAPHIQUE 

La figure « Matriochka » ci-dessous, illustre les imbrications des 2 types possibles de graphique (chart)  d’EXCEL :
[image: ]

Dans EXCEL, un objet Graphique (chart) peut résider :
· Dans une feuille de données (worksheet) par exemple, celle contenant les données sources du graphique (ou dans une autre feuille de données du classeur). Nous parlerons alors de « graphique incorporé » (embedded chart). Une feuille de données peut contenir plusieurs graphiques incorporés.
Un graphique incorporé est membre de la collection « ChartObjects » de l’objet « Worksheet ».
· Dans une feuille graphique (chart sheet). Nous parlerons alors de graphique isolé (standalone chart). Une feuille graphique ne peut contenir qu’un seul graphique isolé épousant tout l’espace de la feuille.
Un graphique isolé est membre de la collection « Charts » de l’objet « Workbook ».
A noter qu’un graphique incorporé possède un « don d’ubiquité » : il est membre de la collection « ChartObjects » et, en même temps, membre de la collection des formes « Shapes ». En conséquence, dans la programmation VBA des graphiques incorporés, nous utilisons les méthodes et propriétés de l’objet « Chart » mais également, certaines méthodes et propriétés de l’objet « Shape ».
[image: ] Pour la manipulation des formes (shapes) par VBA, reportons-nous au document : 
         https://www.excel-pratique.com/fr/telechargements/doc-vba/manipuler-formes-vba-no484

[bookmark: _Toc46148470]Bouclons sur les graphiques contenus dans un classeur

Le fragment de code suivant énumère dans la fenêtre d’exécution le nom de tous les graphiques (incorporés ou isolés) présents dans le classeur :
[image: ] 



[image: ] Pour afficher la fenêtre d’exécution : Ctrl+G. Reportons-nous au document : 
         https://www.excel-pratique.com/fr/telechargements/doc-vba/debogage-no469


[bookmark: _Toc46148471]CREONS UN NOUVEAU GRAPHIQUE (CHART)

[bookmark: _Toc46148472]CREONS UN GRAPHIQUE INCORPORE
Nous créons un graphique incorporé à l’aide de l’une ou l’autre des 2 méthodes suivantes :
[bookmark: _Toc46148473]Méthode  ChartObjects.Add (Left, Top, Width, Height)
Cette méthode insère dans la feuille un graphique incorporé du type défini par défaut, dont la position est déterminée par les paramètres ‘Left (gauche)’ et ‘Top (haut)’ exprimés en points et la taille par les paramètres ‘Width (largeur)’ et ‘Height (hauteur)’ également exprimés en points.

Le fragment de code ci-dessous créé un nouveau graphique incorporé du type défini par défaut :Sub AddDefaultChart()
    Dim oSheet As Worksheet
    Dim oChartObject As ChartObject
    Dim oChart As Chart    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets(1)
    'On créé un nouveau graphique incorporé du type par défaut
    Set oChartObject = oSheet.ChartObjects.Add(600, 20, 300, 200)
    ‘On réfère le graphique (pour utilisation ultérieure)
    Set oChart = oChartObject.Chart    
End Sub



	
[image: ] Pour définir ou modifier le type de graphique par défaut, reportons-nous à la page :
https://fr.officetooltips.com/excel_2016/tips/comment_changer_le_graphique_par_defaut.html)



[bookmark: _Toc46148474]Méthode  Shapes.AddChart2 (style, XlChartType, Left, Top, Width, Height, NewLayout) 

Cette méthode, plus ‘riche’ que la précédente insère dans la feuille un graphique incorporé dont on définit le style et le type, ainsi que la position déterminée par les paramètres ‘Left (gauche)’ et ‘Top (haut)’ exprimés en points et la taille par les paramètres ‘Width (largeur)’ et ‘Height (hauteur)’ également exprimés en points. Le paramètre NewLayout indique que le titre et les légendes du graphique sont automatiquement activés lorsqu’il est valorisé à « true ». 
Le fragment de code ci-dessous créé un nouveau graphique incorporé de type « Histogramme groupé » dans le style 12 avec titre et légendes :Sub AddTypedChart()
    Dim oSheet As Worksheet
    Dim oShape As Shape
    Dim oChart As Chart
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets(1)
    'On créé un nouveau graphique incorporé d'un style et d’un type déterminés
    Set oShape = oSheet.Shapes.AddChart2(12, xlColumnClustered, 600, 20, 600, 460, True)
    ‘On réfère le graphique (pour utilisation ultérieure)
    Set oChart = oShape.Chart
End Sub

  

[image: ]  Le paramétrage de la méthode « Shapes.AddChart2 » comporte les particularités suivantes :
· Lorsque nous voulons activer le style par défaut du xlChartType indiqué, nous valorisons le paramètre « Style » à « -1 ».
Sinon, il n’est pas évident de connaitre la valeur désirée du paramètre « Style ». Le plus simple pour la déterminer consiste à utiliser l’enregistreur de macros.

Pour une bonne utilisation de l’enregistreur de macros, reportons-nous au document : https://www.excel-pratique.com/fr/telechargements/doc-vba/enregistreur-macro-excel-no467.

· Pour déterminer la valeur du paramètre « xlChartType », reportons-nous à l’énumération : 
https://docs.microsoft.com/fr-fr/office/vba/api/excel.xlcharttype




[bookmark: _Toc46148475]CREONS UN GRAPHIQUE ISOLE
[bookmark: _Toc46148476]Méthode  Charts.Add2 (Before, After, Count, NewLayout).
· Les paramètres « Before (avant) » et « After (après) » permettent de positionner la nouvelle feuille graphique dans le classeur. Nous valorisons un seul de ces 2 paramètres en référençant une feuille préexistante du classeur (facultatif).
· Le paramètre « Count (nombre) » permet d’indiquer le nombre de feuilles graphiques identiques à créer (par défaut = 1).
· Le paramètre NewLayout indique que le titre et les légendes du graphique sont automatiquement activés lorsqu’il est valorisé à « true ». 
Le fragment de code suivant créé une nouvelle feuille graphique du type défini par défaut, positionné comme première feuille du classeur, avec titre et légendes et ayant pour nom « ComparatifDépenses » :Sub Add2StandaloneChart()
    Dim oChart As Chart
    
    'On créé une nouvelle feuille graphique du type par défaut positionnée comme première feuille du classeur
    Set oChart = ThisWorkbook.Charts.Add2(ThisWorkbook.Sheets(1), , , True)
    With oChart
        'On nomme la nouvelle feuille graphique
        .Name = "ComparatifDepenses"
    End With
End Sub


[image: ]Le nom d’une feuille graphique doit être unique dans le classeur. En cas de doublon, VBA retourne une erreur.


	
[image: ]  La méthode Chart.Add () utilisée dans les versions anciennes d’EXCEL peut encore être utilisée bien qu’elle ne soit plus ni documentée ni exposée. La raison de cette survivance tient vraisemblablement aux nécessités de compatibilité des solutions développées dans le passé.
Pour les solutions réalisées avec les versions actuelles d’EXCEL, l’utilisation de cette méthode n’est pas recommandée.

Dans la suite de ce document, afin d’illustrer les éléments spécifiés, nous utilisons un classeur type composé de la façon suivante :
· Dans la feuille « Données » du classeur, est défini un tableau structuré contenant des dépenses (fictives) par postes budgétaires (fictifs) :
	Poste
	janv-20
	févr-20
	mars-20
	avr-20

	AUTEUR
	    3 453,27 € 
	       5 000,00 € 
	3 453,27 €
	     3 453,27 € 

	CACHETS
	    4 828,34 € 
	       7 000,00 € 
	      4 828,34 € 
	     1 000,00 € 

	DECOR
	    5 017,55 € 
	       1 000,00 € 
	      5 017,55 € 
	         500,00 € 

	DEPLACEMENTS
	    2 518,19 € 
	       3 000,00 € 
	      2 518,19 € 
	     6 000,00 € 

	DIVERS
	    9 698,05 € 
	       5 000,00 € 
	      9 698,05 € 
	         300,00 € 

	DROITS
	    2 530,21 € 
	       5 000,00 € 
	      2 530,21 € 
	     7 000,00 € 

	FRAIS FIXES
	    4 750,71 € 
	       2 000,00 € 
	      4 750,71 € 
	     2 000,00 € 

	FRAIS POSTAUX
	    2 251,13 € 
	          500,00 € 
	      2 251,13 € 
	     5 000,00 € 

	HONORAIRES
	    4 217,33 € 
	       2 000,00 € 
	      4 217,33 € 
	     2 500,00 € 

	Total
	  39 264,77 € 
	    30 500,00 € 
	    39 264,77 € 
	   27 753,27 € 



· Grâce au code suivant, est créé un graphique incorporé illustrant les dépenses pour le mois de janvier 2020:Sub AddSampleChart()
    Dim oSheet As Worksheet
    Dim oChartObject As ChartObject
    Dim oChart As Chart
    Dim oSerie As Series
    
    'On référe la feuille de destination du graphique contenant les données
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On créé un nouveau graphique incorporé du type par défaut
    Set oChartObject = oSheet.ChartObjects.Add(1, 200, 300, 200)
    'On réfère le graphique créé
    Set oChart = oChartObject.Chart
    'On ajoute une série au graphique
    Set oSerie = oChart.SeriesCollection.NewSeries
    With oSerie
        'On indique la plage utilisée pour les valeurs de la série
        .Values = oSheet.Range("B3:B11")
        'On indique la plage utilisée pour les libellés de l'axe horizontal
        .XValues = oSheet.Range("A3:A11")
    End With
End Sub


Résultat obtenu :
[image: ]
[image: ]  Dans la suite de ce document, les fragments de code proposés manipulent le graphique créé dans le classeur exemple. Ce graphique est de type graphique incorporé.
Au cas où l’on manipule un graphique isolé, la plupart des codes VBA proposés fonctionnent en remplaçant simplement les lignes de référencement de l’objet Chart :
Référencement de l’objet Chart pour un graphique incorporé :
    Dim oSheet As Worksheet
    Dim oChart As Chart
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart

Référencement de l’objet Chart pour un graphique isolé :
    Dim oChart As Chart    
    'On réfère le graphique isolé
    Set oChart = ThisWorkbook.Charts(1)

[bookmark: _Toc46148478]SPECIFIONS LES DEUX ZONES DE L’OBJET CHART

[image: ]


[bookmark: _Toc46148479]La propriété Chart.ChartArea représente la zone du graphique dans son intégralité.
 Le code suivant modifie certains attributs de la zone de graphique de notre exemple :Sub ModifyChartAreaProperties()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oChartArea As ChartArea
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la zone de graphique
    Set oChartArea = oChart.ChartArea
    With oChartArea
        'On modifie la hauteur , la position gauche et la forme de la bordure extérieure de la zone de graphique
        .Height = 400
        .Width = 600
        .RoundedCorners = True
    End With
End Sub

Résultat obtenu :

[image: ]


[image: ]Notons que l’objet Format de l’objet ChartArea d’un graphique incorporé est en lecture seule.
Pour modifier, la couleur de fond du graphique, nous utilisons l’objet parent de l’objet Chart (i.e. ChartObject). Cf. le code ci-dessous.
L’enregistreur de macro privilégie, lui, l’utilisation de l’objet Shape sous-jacent à l’objet Chart pour obtenir le même résultat.
Pour obtenir les paramètres de la fonction RGB() correspondants à la couleur désirée, reporterons-nous à la page :
https://www.excel-pratique.com/fr/vba/liste-couleurs-rgb


Le code suivant modifie la couleur de fond de notre exemple :
Résultat obtenu :Sub modifyChartAreaBackcolor()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oChartArea As ChartArea
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'L'objet Format de ChartArea est en lecture seule -> on passe par l'objet parent (ChartObject) 
     oChart.Parent.Interior.Color = RGB(224, 238, 224)
End Sub

[image: ]


[bookmark: _Toc46148480]La propriété Chart.PlotArea représente la zone de traçage du graphique.
La zone de traçage est incluse dans la zone de graphique.
Le code suivant modifie la taille et la couleur de fond de la zone de traçage de notre exemple :Sub ModifyPlotAreaProperties()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oPlotArea As PlotArea
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la zone de traçage
    Set oPlotArea = oChart.PlotArea
    With oPlotArea
        'On modifie certaines propriétés de la zone de traçage : hauteur et largeur à 75% de la zone de graphique
        .InsideHeight = oChart.ChartArea.Height * 0.75
        .InsideWidth = oChart.ChartArea.Width * 0.75
        'On efface les éventuels formats de la zone de traçage
        .ClearFormats
        'On affecte les caractéristiques du fond de la zone de traçage
        With .Format.Fill
            'On affecte un dégradé depuis le coin
            .OneColorGradient msoGradientFromCorner, 1, 1
            'On affecte la couleur grise estompée
            .BackColor.RGB = RGB(153, 153, 153)
        End With
    End With
End Sub




Résultat obtenu :

[image: ]

[image: ]Notons que, si nous affectons à la propriété InsideHeight de l’objet PlotArea une valeur telle que la zone de traçage de retrouve en dehors de la zone du graphique, EXCEL modifiera automatiquement la valeur pour faire en sorte que la zone de traçage reste incluse dans la zone de graphique.
De même pour les propriétés InsideWidth, InsideLeft, InsideTop.

[image: ]


[bookmark: _Toc46148482]L’objet ChartTitle représente le titre du graphique.
Le code suivant affecte et met en forme le titre de notre exemple : Sub SetChartTitle()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oChartTitle As ChartTitle
    Dim lPos As Long
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On indique que le graphique possède un titre
    oChart.HasTitle = True
    'On réfère le titre
    Set oChartTitle = oChart.ChartTitle
    With oChartTitle
        'On compose le texte du titre composé de 2 lignes
        .Text = "Dépenses mensuelles" & vbCr & "par postes budgétaires"
        'On récupère la position du saut de ligne
        lPos = InStr(1, .Text, vbCr)
        'On modifie les caractéristiques de la première ligne du titre
        With .Characters(1, lPos - 1)
            .Font.Size = 12
            .Font.Bold = True
            .Font.Color = vbBlue
        End With
        'On modifie les caractéristiques de la deuxième ligne du titre
        With .Characters(lPos)
            .Font.Size = 10
            .Font.Bold = False
            .Font.Italic = True
            .Font.Color = vbYellow
        End With
    End With
End Sub



Résultat obtenu :

[image: ]

 
[image: ] Dans la version EXCEL 365 (et antérieures), la longueur du texte du titre d’un graphique est limitée à 255 caractères.
Si nous affectons un texte dont la longueur dépasse 255 caractères, VBA retourne une erreur à l’exécution




Remarquant que le titre inséré par le code précédent se trouve à l’intérieur de la zone de traçage, nous exécutons le code suivant afin d’isoler le titre de notre exemple : Sub IsoleChartTitle()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oPlotArea As PlotArea
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la zone de traçage
    Set oPlotArea = oChart.PlotArea

    oPlotArea.InsideTop = oChart.ChartArea.Top - 20
End Sub

Résultat obtenu :

[image: ]


[bookmark: _Toc46148483]L’objet Legend représente le cartouche de légende du graphique.
Le code suivant déplace la légende de notre exemple en bas du graphique et le borde : Sub SetChartLegend()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oChartLegend As Legend
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On indique que le graphique possède un cartouche de légende
    oChart.HasLegend = True
    'On réfère le cartouche
    Set oChartLegend = oChart.Legend
    With oChartLegend
        'On déplace la légende en bas du graphique
        .Position = xlLegendPositionBottom
        'On formate la bordure
        With .Border
            .Color = vbBlack
            .LineStyle = xlDashDot
            .Weight = xlThin
        End With
    End With
End Sub



Résultat obtenu :

[image: ]



Nous remarquons que la légende déplacée par le code précédent est trop proche des libellés de l’axe des abscisses.  Nous exécutons le code suivant afin d’isoler la légende de notre exemple : Sub IsoleLegend()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oPlotArea As PlotArea
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la zone de traçage
    Set oPlotArea = oChart.PlotArea
    
    oPlotArea.InsideHeight = oPlotArea.InsideHeight * 0.85
End Sub

Résultat obtenu :

[image: ]

[bookmark: _Toc46148484]L’objet Axis désigne l’un des axes du graphique.
Nous utilisons la collection Axes (xlAxisType) pour référencer un objet Axis.
L’énumération xlAxisType contient les constantes suivantes :
· xlCategory pour désigner l’axe des abscisses.
· xlValue pour désigner l’axe des ordonnées.
· xlSeriesAxis désigne l’axe de la 3ième dimension d’un graphique en 3D.
Le code suivant donne un titre à l’axe des abscisses de l’exemple, ainsi que colorise les légendes et le titre de la même couleur :Sub SetAbscisses()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oAxis As Axis
    Dim lgvsViolet As Long
    
    'On valorise une variable locale à la couleur violet
    lgvsViolet = RGB(112, 48, 160)
   
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère l'axe des abscisses
    Set oAxis = oChart.Axes(xlCategory)
    With oAxis
        'On indique que l'axe a un titre
        .HasTitle = True
        With .AxisTitle
            'On indique le titre de l'axe
            .Text = "Postes budgétaires"
            'On formate et colorise le titre de l'axe
            .Font.Size = 10
            .Font.Italic = True
            .Font.Color = lgvsViolet
        End With
        'On colorise les étiquettes
        .TickLabels.Font.Color = lgvsViolet
    End With
End Sub



Résultat obtenu :

[image: ]



Le code suivant donne un titre à l’axe des ordonnées de l’exemple, y ajoute une flèche et colorise l’axe, les légendes et le titre de la même couleur :Sub SetOrdonnees()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oAxis As Axis
    Dim lgvsViolet As Long
    
    'On valorise une variable locale à la couleur violet
    lgvsViolet = RGB(112, 48, 160)
   
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère l'axe des ordonnées
    Set oAxis = oChart.Axes(xlValue)
    With oAxis
        With .Format.Line
            'On positionne une flêche sur l'axe des ordonnées
            .EndArrowheadStyle = msoArrowheadTriangle
            'On fixe la taille de l'axe
            .Weight = 1.5
            'On colorise l'axe des ordonnées en bleu
            .ForeColor.RGB = RGB(0, 0, 255)
            'On indique que l'axe a un titre
        End With
        'On indique que l'axe a un titre
        .HasTitle = True
        With .AxisTitle
            'On indique le titre de l'axe
            .Text = "Dépenses"
            'On formate et colorise le titre de l'axe
            .Font.Size = 10
            .Font.Italic = True
            .Font.Color = vbBlue
        End With
        'On colorise les étiquettes
        .TickLabels.Font.Color = vbBlue
    End With
End Sub



Résultat obtenu :

[image: ]



[bookmark: _Toc46148485]SPECIFIONS LA COLLECTION SERIESCOLLECTION

La collection SeriesCollection rassemble toutes les séries de valeurs présentes dans le graphique.
L’objet Series désigne une série particulière de données associées au graphique.
Nous affectons l’objet Series en référençant un élément particulier de SeriesCollection par son index.
L’objet Series possède une collection DataLabels rassemblant les étiquettes de données de la série.
Le code suivant ajoute et formate des étiquettes de données sur la série 1 :Sub addDatalabels()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oSerie As Series
    Dim oDatalabel As DataLabel
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la première série
    Set oSerie = oChart.SeriesCollection(1)
    'On indique que la série possède des étiquettes
    oSerie.HasDataLabels = True
    For Each oDatalabel In oSerie.DataLabels
        'On colorise le texte de l'étiquette et la met en italique
        With oDatalabel
            .Font.Color = vbBlue
            .Font.Italic = True
        End With
    Next
End Sub



Résultat obtenu :

[bookmark: _Toc46148486]AJOUTONS DES SERIES DE DONNEES AU GRAPHIQUE

[bookmark: _Toc46148487]La méthode SeriesCollection.NewSeries nous permet d’ajouter une nouvelle série de données au graphique.
Le code suivant ajoute 3 nouvelles séries à notre exemple reprenant les valeurs des postes budgétaires pour les mois de février, mars et avril 2020 :Sub AddSeries()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oSerie As Series
    Dim i As Integer
    
    'On réfère la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On réfère le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réalise une boucle de 3 itérations
    For i = 1 To 3
        'On ajoute une série de données
        Set oSerie = oChart.SeriesCollection.NewSeries
        'On référençe la plage de valeurs de la colonne correspondante du tableau de données
        oSerie.Values = oSheet.Range("B3:B11").Offset(, i)
    Next
End Sub

Résultat obtenu :
[image: ]
[bookmark: _Toc46148488]MODIFIONS LES CARACTERISTIQUES DES SERIES

[bookmark: _Toc46148489]Les objets Series.Values et Series.XValue nous permettent d’indiquer respectivement la plage de valeurs de la série et la plage des libellés de l’axe des abscisses.
Le code utilisé pour initier notre exemple spécifie les valeurs et étiquettes de la série 1 du graphique :
Sub AddSampleChart()
    Dim oSheet As Worksheet
    Dim oChartObject As ChartObject
    Dim oChart As Chart
    Dim oSerie As Series
    
    'On référe la feuille de destination du graphique contenant les données
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On créé un nouveau graphique incorporé du type par défaut
    Set oChartObject = oSheet.ChartObjects.Add(1, 200, 300, 200)
    'On réfère le graphique créé
    Set oChart = oChartObject.Chart
    'On ajoute une série au graphique
    Set oSerie = oChart.SeriesCollection.NewSeries
    With oSerie
        'On indique la plage utilisée pour les valeurs de la série
        .Values = oSheet.Range("B3:B11")
        'On indique la plage utilisée pour les libellés de l'axe horizontal
        .XValues = oSheet.Range("A3:A11")
    End With
End Sub


[bookmark: _Toc46148490]L’objet Series.Format nous permet d’agir sur les caractéristiques de l’objet Serie.
Dans notre exemple, du fait que l’ajout des séries réalisé avec le code précédent, laisse le soin à EXCEL d’affecter les couleurs aux barres, la série 3 en gris tend à se confondre avec le fond de la zone de traçage. Pour plus de lisibilité, nous désirons modifier la couleur des barres de la série 3. Le code suivant a cette fonction :Sub changeBarColor()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oSerie As Series
    Dim oFormat As ChartFormat
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On réfère la troisième série
    Set oSerie = oChart.SeriesCollection(3)
    'On réfère le format de la série
    Set oFormat = oSerie.Format
    'On modifie la couleur de la barre
    oFormat.Fill.ForeColor.RGB = RGB(153, 102, 0)
End Sub


Résultat obtenu :

[image: ]



[bookmark: _Toc46148491]La propriété Series.Name nous permet de donner un nom aux séries.
Remarquons que la légende d’un graphique reprend le nom des séries (et leur couleur).Sub SetSeriesName()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oSerie As Series
    Dim i As Integer
    
    'On réfère la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On réfère le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    For i = 1 To 4
        Set oSerie = oChart.SeriesCollection(i)
        'On affecte le nom de la série par référence à l'entête de la colonne du tableau
        oSerie.Name = oSheet.Range("A2").Offset(, i).Value
    Next    
End Sub

Le code suivant affecte à chaque série un nom répliquant les entêtes des colonnes du tableau de données :

Résultat obtenu :


[image: ]


[bookmark: _Toc46148492]AJOUTONS UNE COURBE DE TENDANCE
La collection TrendLines d’un objet Serie rassemble toutes les courbes de tendance définies pour une série.
L’objet TrendLine désigne une courbe de tendance particulière d’une série donnée.
[bookmark: _Toc46148493]Méthode Serie.TrendLines.Add(xlTrendLineType,…) permet d’ajouter un nouvel objet TrendLine 
Le code suivant ajoute une courbe de tendance sur la moyenne de 2 valeurs successives de la série 1 :Sub AddTrendLine()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    Dim oTendance As Trendline
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On ajoute une courbe de tendance sur la moyenne entre 2 valeurs de la série 1
    Set oTendance = oChart.SeriesCollection(1).Trendlines.Add(xlMovingAvg)
    'On formate la courbe de tendance : couleur rouge, épaisseur 1.5
    With oTendance.Format.Line
        .Visible = msoTrue
        .ForeColor.RGB = RGB(255, 0, 0)
        .Weight = 1.5
    End With
End Sub

(notons que dans notre exemple, ce type de courbe de tendance n’a pas grand sens…)
Résultat obtenu :

[image: ]
[bookmark: _Toc46148494]APPLIQUONS UN STYLE PREDEFINI AU GRAPHIQUE
Nous avons vu dans les paragraphes précédents qu’il nous est loisible de modifier individuellement les caractéristiques d’un graphique. 
EXCEL offre également la possibilité d’appliquer un « ChartStyle » prédéfini à un graphique existant. Nous modifierons ainsi en une seule fois plusieurs des caractéristiques du graphique.
[bookmark: _Toc46148495]La propriété Chart.Style (lStyle) nous permet d’appliquer le style désiré.
Pour déterminer la valeur du paramètre « lStyle», le plus simple consiste à utiliser l’enregistreur de macros en simulant manuellement la transformation désirée.
Le code suivant transforme notre graphique en un graphique en 3D sur fond noir :Sub changeChartStyle()
    Dim oSheet As Worksheet
    Dim oChart As Chart
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On rétablit la mise en forme par défaut du graphique
    oChart.ClearToMatchStyle
    'On applique le style désiré : Barres 3D fond noir
    oChart.ChartStyle = 294
End Sub

Résultat obtenu :

[image: ]
[bookmark: _Toc46148496]SPECIFIONS QUELQUES METHODES UTILES DE LA CLASSE CHART

[bookmark: _Toc46148497]EXPORTONS UN GRAPHIQUE EN PDF
[bookmark: _Toc46148498]Méthode Chart.ExportAsFixedFormat (Type, FileName,…)
Le code suivant exporte notre graphique exemple dans un fichier PDF :Private Sub exportChart()
    Dim sMess As String
    Dim lElementID As Long, lArg1 As Long, lArg2 As Long
    Dim oSheet As Worksheet
    Dim oChart As Chart
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On exporte le graphique dans un fichier PDF
    oChart.ExportAsFixedFormat ExportAsFixedFormat, ThisWorkbook.Path & "\" & "EXCEL_PRATIQUE_BP_CHART"

End Sub


[bookmark: _Toc46148499]EXPORTONS UN GRAPHIQUE AU FORMAT JPEG
[bookmark: _Toc46148500]Méthode Chart.Export (FileName…)
Le code suivant enregistre notre graphique exemple dans un nouveau fichier image au format JPEG :Private Sub exportToJPEG()
    Dim sMess As String
    Dim sMess As String
    Dim lElementID As Long, lArg1 As Long, lArg2 As Long
    Dim oSheet As Worksheet
    Dim oChart As Chart
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On exporte le graphique au format image JPEG
    oChart.Export ThisWorkbook.Path & "\" & "EXCEL_PRATIQUE_BP_CHART.jpg"    
End Sub


[bookmark: _Toc46148501]IMPRIMONS UN GRAPHIQUE
[bookmark: _Toc46148502]Méthode Chart.PrintOut (…)
Le code suivant envoie notre graphique exemple vers l’imprimante par défaut :Private Sub printChart()
    Dim sMess As String
    Dim lElementID As Long, lArg1 As Long, lArg2 As Long
    Dim oSheet As Worksheet
    Dim oChart As Chart
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On imprime le graphique
    oChart.PrintOut
End Sub


[bookmark: _Toc46148503]ENREGISTRONS UN GRAPHIQUE EN TANT QUE MODELE
[bookmark: _Toc46148504]Méthode Chart.SaveChartTemplate(Filename)
Le code suivant enregistre notre graphique exemple en tant que modèle « DepensesParPostes.crtx » :Private Sub printChart()
    Dim sMess As String
    Dim lElementID As Long, lArg1 As Long, lArg2 As Long
    Dim oSheet As Worksheet
    Dim oChart As Chart
    
    'On référe la feuille
    Set oSheet = ThisWorkbook.Worksheets("Données")
    'On référe le graphique incorporé
    Set oChart = oSheet.ChartObjects(1).Chart
    'On exporte le graphique sous forme de graphique modèle
    oChart.SaveChartTemplate ThisWorkbook.Path & "\" & "DepensesParPostes"
End Sub

Sub ListerGraphs ()
Dim oSheet As Object
Dim oChartObject As ChartObject
Dim oChartobjects As ChartObjects

'on boucle sur les objets sheet du classeur
For Each oSheet In ThisWorkbook.Sheets
'on teste la nature de la feuille
Select Case TypeName (oSheet)
'Dans le cas oi la feuille est une feuille de données
Case Is = "Worksheet"
'on boucle sur les objets ChartObject de la feuille
For Each oChartObject In oSheet.ChartObjects
'on affiche le nom du graphique dans la fenétre d'exécution
Debug.Print "Graphique incorporé : " & oChartObject.Name

'Dans le cas oi la feuille est une feuille graphique

End Select
Next
End Sub

